
 

Décembre 2013 - 1 - Support Clients SAS France 

NOUVEAUTÉS ODS LAYOUT EN SAS 9.4 

 
Expérimentales depuis SAS 9, les fonctionnalités ODS 
LAYOUT et RWI passent en statut « production » avec 
SAS 9.4 et apportent une puissance de production de 
rapports accrue. 

1. OBJECTIFS 

L’ODS LAYOUT et son complément « RWI – Report Writing Interface » permettent de créer des 
rapports personnalisés combinant tous types de sorties produites par SAS Foundation. 
Ces fonctionnalités, apparues avec SAS 9, étaient jusqu’à SAS 9.3 en statut « pré-production », ce 

qui signifie fournies en l’état, avec un support limité et soumises à changements.  
Avec SAS 9.4, elles passent en statut « production » et sont enrichies de nouvelles fonctionnalités. 

 
Cet article va décrire le principe de ces fonctionnalités et en présenter les nouveautés apportées par 
SAS 9.4. 
 

2. PRINCIPE DE L’ODS LAYOUT ET DE RWI 

2.1. Concept de l’ODS LAYOUT 

L’ODS LAYOUT permet d’arranger des graphiques, du texte, des tableaux sur une page « virtuelle » 
avec toute la précision requise – allant jusqu’au pixel ! 
Le schéma suivant illustre son principe et sa terminologie : 

 Caractéristiques : 

Catégories : Base, ODS 
OS : Windows, Unix, z/OS 
Version : SAS® 9.4 
Vérifié en décembre 2013 

 La page virtuelle du rapport 

contient des titres (Titles) et des 
notes de bas de page (Footnotes). 
 

 Le « Layout container » ne peut 
contenir qu’un ensemble de 
régions (« region container »). 

 
 Chaque région contient les 

résultats de sortie SAS (texte, 
graphique, image, tableaux…) ou 
des « Layout container » 
imbriqués apportant une 

souplesse quasi infinie. 
 

 Chaque « Layout container » ou 

région peut avoir une taille fixe ou 
dynamique. 

 
ODS LAYOUT se décline sous deux formes – chacune ayant ses avantages et ses contraintes : 

 « Absolute » plus utilisée pour des rapports statiques 

 « Gridded » plus adaptée aux rapports dynamiques. La zone de LAYOUT peut se comparer à 
une feuille d’un classeur Excel (grille de colonnes et de lignes, chaque région étant une cellule 
de la grille) 

 
Son utilisation suit la logique du système ODS à savoir que ses instructions encadrent le code SAS 
des procédures comme suit : 

ods layout… ; 

code SAS produisant des sorties 
ods layout end ; 

 
L’imbrication des « Layout container » au sein des régions offre une souplesse d’emploi quasi illimitée 
d’autant que les deux formes de LAYOUT « Absolute » et « Gridded » peuvent être combinées : un 
« Layout container Absolute » peut contenir une région qui va elle-même contenir un « Layout 
container Gridded » et réciproquement. Le choix du type de LAYOUT « Absolute » ou « Gridded » 

dépend du type de rapport à produire. 
 
L’ODS LAYOUT peut s’appliquer sur des formats de sortie HTML, PRINTER (inclut PDF, PCL, PSL) et 
« rapport SAS ». Pour les formats de sortie de type PRINTER, la notion de page « virtuelle » doit 

prendre en compte la contrainte  - et donc les limites - de la page physique qui sera le support du 
rapport.  

Les options de pagination se définissent via les options globales du langage SAS : 
 PAPERSIZE    Taille de la page 
 ORIENTATION    Orientation de la page (portrait…) 
 TOPMARGIN, BOTTOMMARGIN, Tailles des marges de la page 

LEFTMARGIN, RIGHTMARGIN 
 

2.2. Concept du Report Writing Interface – RWI 

Le RWI est l’équivalent « programmatique » de l’ODS LAYOUT couplé à la puissance du langage SAS. 
RWI permet d’aller au-delà des rapports de type « tabulaires ». 
Communément appelé production de rapports via une étape DATA sans données (« DATA _NULL_ 
report writing »), il permet d’exécuter toutes les instructions d’ODS LAYOUT (« Layout container » 

avec des régions) via un langage objet utilisé au sein d’une étape « DATA » SAS. Ainsi il combine le 

meilleur des 2 mondes : 

 La puissance programmatique de l’étape DATA : accès aux données, filtre et logique 

conditionnelle, regroupements, formats… 
 La richesse de l’ODS par le positionnement et la mise en forme des objets : textes, images, 

graphiques… 
Dans les exemples de cet article nous verrons que chaque instruction ODS LAYOUT a son équivalent 
programmatique RWI. 
 

3. CARACTÉRISTIQUES ET NOUVEAUTÉS 

3.1. Titres et notes de bas de page 

Le comportement des titres (TITLE) et notes de bas de page (FOOTNOTE) a été modifié avec SAS 
9.4 et ODS LAYOUT. 
Le comportement traditionnel quand il n’y a pas de LAYOUT reste inchangé, avant l’exécution des 
procédures SAS, les titres et notes de bas de page sont placés respectivement en haut et bas de 

page et sont répétés sur chaque page jusqu’à la fin du rapport, sauf option spécifique. 
 
Avec la présence d’instructions LAYOUT, si la première instruction pour alimenter la page est de type 
LAYOUT, le comportement reste inchangé comme décrit ci-dessus. Par contre si une instruction 
LAYOUT est rencontrée au cours de la page, seuls les titres et notes de bas de page ayant changé 
sont affichés avant la zone de LAYOUT. De manière identique, quand une instruction de REGION est 
exécutée au sein du LAYOUT, seuls les titres et notes de bas de page ayant changé sont affichés 

avant la REGION. Le même principe s’applique si la REGION contient plusieurs procédures (PROC) 
avec des titres et notes de bas de pages, seuls ceux ayant changé seront affichés respectivement 
autour de leur procédure. 
 

3.2. Positionnement par coordonnées X Y 

Avant SAS 9.4, seul le LAYOUT ABSOLUTE permettait de spécifier des coordonnées de 
positionnement sur la page sous forme : 

 X abscisse, donnant la position de départ sur l’axe horizontal, l’origine du repère étant en 
haut et à gauche de la page, 

 Y ordonnée, donnant la position de départ sur l’axe vertical (orienté du haut vers le bas). 
 
Avec SAS 9.4 les coordonnées X et Y sont valides aussi en LAYOUT GRIDDED. 

Ces coordonnées peuvent être exprimées en diverses unités : 
 in  pouce (défaut) 
 cm  centimètre 
 mm millimètre 
 % pourcentage par rapport à la dimension de la page 
 px pixel 

 pt point (valant 1/72 de pouce) 
 

Ces coordonnées ne sont supportées que pour des destinations PRINTER (inclut PDF, PCL, PSL). 
 

3.3. OPTION STYLE 

L’option « STYLE= » peut être indiquée soit dans une instruction « ODS LAYOUT ;» ou une instruction 
« ODS REGION ;» et permet d’enrichir le style en cours ou de surcharger (remplacer ponctuellement) 
un de ses attributs. 
La syntaxe est : 
STYLE = <élément de style> [<attribut de style>] 
 
Attention : cette fonctionnalité est cependant considérée pré-production en SAS 9.4. 

 

3.4. PROC ODSLIST et PROC ODSTEXT 

Nouveautés de SAS 9.4, ces procédures permettent de créer des listes à puces et des blocs de texte 
dans les rapports. Ces listes et blocs de texte peuvent contenir des données statiques ou dynamiques 

(provenant de tables). PROC ODSTEXT offre plus de possibilités et de lisibilité que l’instruction « ODS 

TEXT=… ». 

3.5. OPTION ADVANCE 

Les rapports actuels sont majoritairement dynamiques et centrés sur les données. Le contenu d’un 
rapport peut inclure la production de tableaux et graphiques de manière itérative. L’ODS LAYOUT 
GRIDDED qui divise la page en cellules (REGION) est plus adapté à ce type de rapport. Il inclut une 
option spécifique appelée ADVANCE qui permet un « peuplement » fin et adapté des cellules. 
ADVANCE peut prendre les valeurs suivantes : 

 EXPLICIT Valeur par défaut, une nouvelle région du LAYOUT est utilisée lorsqu’une 
instruction « ODS REGION ; » est spécifiée. 

 PROC Changement de région chaque fois qu’une nouvelle procédure (PROC) est 
spécifiée – à moins qu’une instruction « ODS REGION ; » soit explicitement 
indiquée avant la procédure. 

 OUTPUT Changement de région chaque fois qu’une sortie graphique ou table est 
rencontrée – à moins qu’une instruction « ODS REGION ; » soit explicitement 

indiquée avant la procédure. 
 BYGROUP Changement de région à chaque fois qu’une procédure (PROC) ou une 

instruction de regroupement (BY) est rencontrée – à moins qu’une instruction 
« ODS REGION ; » soit explicitement indiquée avant la procédure. Si la 
procédure ne contient pas de regroupement, ODS avance jusqu’à la procédure 
suivante. Ainsi pour des procédures avec ou sans regroupement, l’option 

BYGROUP garantit un changement de région à chaque nouvelle procédure. 
 

3.6. Peuplement des régions en LAYOUT GRIDDED 

En LAYOUT GRIDDED, la page du rapport est divisée en un tableau à deux dimensions, chaque cellule 
constituant une région. Le peuplement des régions se définit par l’option ORDER_TYPE et peut se 

faire de 2 manières différentes : 
 Ligne par ligne, soit horizontalement, cela suppose un nombre fixe de colonnes. Dans ce cas 

ORDER_TYPE peut être spécifié à ROW_MAJOR (c’est la valeur par défaut). Cette option est 
valide pour des formats de sortie HTML, PRINTER (inclut PDF, PCL, PSL). 

 Colonne par colonne, soit verticalement, cela suppose un nombre fixe de lignes. Dans ce cas 
ORDER_TYPE doit être spécifié à COLUMN_MAJOR. Cette option est valide pour des formats 
de sortie PRINTER (inclut PDF, PCL, PSL). 

 

4. EXEMPLES 

Les exemples illustrés dans ce paragraphe sont téléchargeables sur notre site. 
 

4.1. Tableau de bord des ventes par trimestre des Cépages d’un 
vignoble 

Nous prenons l’exemple d’une exploitation viticole et la production d’un rapport qui va donner un 
graphique par trimestre des ventes de chacun des cépages. Cet exemple va illustrer la puissance de 
l’option ADVANCE de l’ODS LAYOUT GRIDDED. 

 

4.1.1. Initialisations et définition de style  

Les options de la page sont définies ainsi qu’un style, avec l’astuce de mettre des marges à « .001in » 
afin d’avoir le style appliqué à toute la page. 
 
options papersize=letter orientation=landscape leftmargin=.001in 

rightmargin=.001in topmargin=.001in bottommargin=.001in nodate nonumber 

nobyline;  

title; footnote; 

proc template; 

  define style Styles.Oakwood;  

    parent = Styles.printer;  

    class Body /  

    backgroundcolor=cxc4d4a1; 

  end;  

run; 

4.1.2. Définition d’un modèle de graphique 

Le modèle de graphique générique permet un affichage de type histogramme par trimestre et sera 
appelé par une PROC SGRENDER.   
proc template;                                                                 

define statgraph Quarterly_vbar / store = SASUSER.TEMPLAT;                  

dynamic QUARTER RESPONSE GROUP FMT TARGET _BYVAL_;                                          

Begingraph;                                                              

  DiscreteAttrVar attrvar=GroupAttrs var=GROUP attrmap="AttrMap"; 

  DiscreteAttrMap name="AttrMap"; 

    Value "Good" / fillattrs=( color=CXC2E699); 

    Value "Satisfactory" / fillattrs=( color=CXFA9FB5); 

    Value "Poor" / fillattrs=( color=CXFF0000); 

  EndDiscreteAttrMap; 

  Entrytitle _BYVAL_;  

  if ( exist(FMT) )  

     Layout overlay / yaxisopts=(display=(ticks tickvalues) 

linearopts=(tickvalueformat=FMT)) xaxisopts=(display=(ticks tickvalues));                                                       

       Barchart x=QUARTER y=RESPONSE / primary=true stat=sum 

group=GroupAttrs groupdisplay=cluster target=TARGET url=URL 

dataskin=pressed;                                                        

     Endlayout; 

  else  

      Layout overlay / yaxisopts=(display=(ticks tickvalues)) 

xaxisopts=(display=(ticks tickvalues));                                                       

       Barchart x=QUARTER y=RESPONSE / primary=true stat=sum 

group=GroupAttrs groupdisplay=cluster target=TARGET url=URL 

dataskin=pressed;                                                        

     Endlayout; 

  endif;  

Endgraph;                                                                

end;                                                                        

run;           

 

4.1.3. Production du rapport 

Le rapport est produit dans un fichier PDF nommé « Revenue.pdf ». Un LAYOUT de type GRIDDED 
est défini. Il est positionné avec des coordonnées X et Y exprimées en pourcentage par rapport à la 
page. Il est découpé en 20 REGIONS (4 lignes et 5 colonnes) de tailles identiques exprimées 
également en pourcentage.  
 
L’option ADVANCE a été mise à BYGROUP. La table PRODUCT contient les ventes trimestrielles de 

chaque cépage produit par l’exploitation viticole.  
 

La Procédure SGRENDER appliquée à la table PRODUCT avec le modèle graphique défini ci-dessus et 
regroupant les produits par cépage, va produire un graphique par cépage qui, grâce à l’option 
ADVANCE mise à BYGROUP, sera ventilé dans une nouvelle région. 
 
ods graphics / noborder; 

 

ods pdf file="Revenue.pdf" notoc style=Styles.Oakwood; 

ods layout gridded x=1.5% y=5% columns=5 column_widths=(18% 18% 18% 18% 

18%)  rows=4 row_heights=(21.5% 21.5% 21.5% 21.5%) advance=bygroup;  
 
 
proc sgrender data=product template=quarterly_vbar ;  

  dynamic QUARTER="quarter" RESPONSE="sales" GROUP="performance" 

FMT="kdollar." TARGET="target_sales"; 

  by product;  

run; 

ods layout end;  

ods _all_ close;  

Voici un extrait du rapport produit : 

 
 

4.2. Rapport journalier des vins disponibles classés par année 

Pour notre exploitation viticole, nous voulons un rapport quotidien des vins disponibles dans le chai 
classés par année. Cet exemple va permettre de montrer l’utilisation de RWI et également de voir 
un rapport centré sur les données – on ne connaît pas le nombre de lignes que produira ce rapport. 
Afin d’avoir un rendu agréable le rapport présentera une photo du chai en partie supérieure gauche 

de la page. 

 

4.2.1. Initialisations 

La table des vins disponibles dans le chai est triée par année décroissante. Les options de pages sont 
définies.  
Le rapport est produit dans un fichier PDF nommé « Cellar.pdf ».  

Un LAYOUT de type GRIDDED est défini. Il est positionné avec juste la coordonnée X exprimée en 
pourcentage par rapport à la page. Il a 2 colonnes et un nombre inconnu de lignes. Sa taille est 
exprimée en pourcentages de la taille de la page. Il utilise l’option de « pré production » STYLE= 
 
proc sort data=wine_cellar reverse;  

  by year; 

run; 

 

title; 

footnote; 

options papersize=letter orientation=portrait nodate nonumber; 

 

ods pdf  file="Cellar.pdf" notoc startpage=never; 

ods layout gridded columns=2 x=5% width=90% height=98% 

style={background=cxc4d4a1}; 

 

4.2.2. Imbrication de LAYOUT 

Les instructions ci-après montrent la puissance de l’ODS LAYOUT.  

 
Une première instruction « ODS REGION ; » avec une largeur de 2,5 pouces occupe la première 
région du LAYOUT (colonne 1, ligne 1).  

 
ods region width=2.5in; 

 
Une seconde instruction « ODS REGION ; » sans indication occupe la seconde région du LAYOUT 
(colonne 2, ligne 1). Puis une nouvelle instruction « ODS LAYOUT ; » indique un LAYOUT imbriqué 
dans la seconde région.  
ods region;  

title "(*ESC*){style [background=cxc4d4a1]Sélections du Chai}"; 

ods layout gridded style={background=cxc4d4a1}; 

 
ODS LAYOUT a un algorithme de calcul de taille de région très robuste. Ainsi il tient compte des 
éléments suivants pour déduire la taille de la seconde région : 

 le LAYOUT parent occupe 90% de la largeur de la page. 

 la première région a une largeur de 2,5 pouces. 

La seconde région a donc une largeur de 90% de la page moins 2,5 pouces. Il est inutile d’ajouter 
d’autres spécifications pour cette taille car cela a pour risque de rendre le code du programme plus 
difficile à maintenir. 

4.2.3. Utilisation de RWI 

Une nouvelle région dans le LAYOUT imbriqué va contenir la liste des vins disponibles de millésime 
inférieur à 2011. L’emploi RWI est judicieux car il utilise dans ce cas le filtre, le regroupement et la 
logique conditionnelle de l’étape DATA pour construire les éléments de la région. 
On remarque les instructions du langage objet pour traiter les éléments ODS. 

Le code est commenté afin de bien suivre le cheminement. 
 
ods region;  

 

data _null_;  

  set wine_cellar(where=(year < 2011)) end=eof;  /* filtre       */ 

   by year;                                      /* regroupement */ 

   if _n_ eq 1 then do;       /* 1ere observation il faut instancier 

l’objet ODS */ 

      dcl odsout obj();  

   end; 

 

   if first.year eq 1 then do; /*Nouvelle annee,titre et debut de 

tableau*/  

 obj.format_text(data: year,  

                 style_elem: "Systemtitles", 

                      style_attr: "backgroundcolor=_undef_");   

 obj.table_start(style_attr: 'frame=void rules=none 

backgroundcolor=_undef_'); 

   end; 

 

   obj.row_start();       /* Ligne courante, vin de l annee a afficher */ 

   obj.format_cell(data: year || ' ' || strip(brand) || ' ' || 

strip(varietal) || ' ' || strip(price) ); 

   obj.row_end();  

 

   if last.year eq 1 then do; /* Fin annee en cours, fin tableau */  

 obj.table_end(); 

   end; 

run; 

ods layout end; /* LAYOUT imbrique */ 

ods layout end; /* LAYOUT parent */ 

 

4.2.4.  Touche finale 

Il reste à positionner la photo du chai pour agrémenter le rapport. Afin de montrer que les LAYOUT 
peuvent être superposés, un nouveau LAYOUT est créé avec des coordonnées X et Y exprimées en 

pouces. 
 
ods layout gridded x=0 y=1in;  

ods region;  

data _null_;  

  dcl odsout obj();  

  obj.image(file: "Images/cellar1.jpeg"); 

;run; 

ods layout end;  

ods _all_ close; 

 
Voici un extrait du rapport produit : 

5. LIENS UTILES 

Cet article est basé sur cette publication – les exemples en sont d’ailleurs extraits. 

 
Un précédent article vous présentait « Report Writing Interface ». Le présent article est focalisé sur 
les nouveautés de SAS 9.4 
 
La documentation sur l’ODS LAYOUT 
 
Tip Sheet : Report Writing Interface - RWI 

 
La documentation sur l’ODS plus généralement. 

 

6. CONCLUSION 

Les fonctionnalités et nouveautés de l’ODS LAYOUT avec SAS 9.4 permettent de produire des 
rapports riches de manière souple et très élaborée tout en restant simples d’utilisation.    
 

 
Jean-Marie POILANE 
Consultant Support Clients SAS France 

